
Introduction to Esterel

I Imperative, textual, concurrent language

I Developed since early 1980s (Gérard Berry)
I Based on synchronous model of time

I Program execution synchronized to an external clock
I Like synchronous digital logic
I Suits the cyclic executive approach

I Same model of computation as SyncCharts/Safe State
Machines (SSMs)

I EsterelStudio generates Esterel from SSMs as intermediate
code

I Currently undergoing IEEE standardization (Esterel v7)

Thanks to Stephen Edwards (http: // www1. cs. columbia. edu/ ~ sedwards/ ) for

providing part of the following material

Introduction to Esterel

Time is divided into discrete ticks (also called cycles, steps,
instants)

Two types of statements:
I Those that take “zero time” (execute and terminate in same

tick, e.g., emit)
I Correspond to Connectors in SSMs

I Those that delay for a prescribed number of ticks (e.g.,
await)

I Correspond to States in SSMs

Signals

I Esterel programs/SSMs communicate through signals
I These are like wires

I Each signal is either present or absent in each tick
I Can’t take multiple values within a tick

I Presence/absence not held between ticks
I Broadcast across the program

I Any process can read or write a signal

Signals

I Status of an input signal is determined by input event, and by
local emissions

I Status of local or output signal is determined per tick
I Default status: absent
I Must execute an “emit S” statement to set signal S present

I await A:
I Waits for A and terminates when A occurs

Synchrony Hypothesis

I Computations are considered to
I take no time
I be atomic

G. Luettgen 2001

Synchronous Model of Computation

To summarize: the synchronous model of computation of
SSMs/Esterel is characterized by:

1. Computations considered to take no time (synchrony
hypothesis)

2. Time is divided into discrete ticks

3. Signals are either present or absent in each tick

Sometimes, “synchrony” refers to just the first two points (e. g., in
the original Statecharts as implemented in Statemate); to explicitly
include the third requirement as well, we also speak of the strict
synchrony

The ABRO Example

I Consider the following controller specification:
I Emit the output O as soon as both the inputs A and B have

been received.
I Reset the behavior whenever the input R is received.

I This is still a bit ambiguous; to complete:
I If R occurs, emit nothing
I Do nothing at initialization time
I Input signals may be simultaneous

The ABRO Example—Mealy Style

S0

S1

S2 S3

S4

ABR/ ABR/

ABR/O

R/

BR/O

R/

AR/O

R/

Write Things Once

I The disadvantage of this (flat) notation:
I Size grows exponentially
I A little change to the specification may incur a major change

to the automaton (often ends with full rewriting)

I The answer:
I Add hierarchy
I More generally: Write Things Once (WTO)

I Analogy from language theory:
I Use regular expressions to represent large (possibly infinite)

sets of strings

ABRO—Safe State Machine

Fall 2007 EE 249 Slide 12



Introduction to Safe State Machines and Esterel
Esterel Language Overview

Esterel/SSM Pragmatics
Interfacing with the Environment

Property Verification

Signals and Synchrony
The ABRO Example
Write Things Once
The multiform notion of time
Uses, Advantages, Disadvantages

ABRO—The Esterel Version

module ABRO:

input A, B, R;

output O;

loop

[ await A || await B ];

emit O

each R

end module

I Esterel programs built
from modules

I Each module has an
interface of input and
output signals

I Much simpler since
language includes notions
of signals, waiting, and
reset

ABRO—The Esterel Version

module ABRO:

input A, B, R;

output O;

loop

[ await A || await B ];

emit O

each R

end module

I loop ... each
statement implements
reset

I || runs the two awaits
in parallel

I await waits for the next
tick where its signal is
present

I Parallel terminates when
all its threads have

I emit O makes signal O
present when it runs

The Multiform Notion of Time

I Some “classical” programming languages already include a
concept of real-time

I Consider the following Ada code fragment, which signals
minutes to a task B:

loop

delay 60;

B.Minute

end

I This works in principle

I However, it is not deterministic!

The Multiform Notion of Time

I A design goal of synchronous languages:
I Fully deterministic behavior
I Applies to functionality and (logical) timing

I Approach:
I Replace notion of physical time with notion of order
I Only consider simultaneity and precedence of events

I Hence, physical time does not play any special role
I Is handled like any other event from program environment
I This is called multiform notion of time

The Multiform Notion of Time

I Consider following requirements:
I “The train must stop within 10 seconds”
I “The train must stop within 100 meters”

I These are conceptually of the same nature!

I In languages where physical time plays particular role, these
requirements are typically expressed completely differently

I In synchronous model, use similar precedence constraints:
I “The event stop must precede the 10th (respectively, 100th)

next occurrence of the event second (respectively, meter)”

The Multiform Notion of Time

I History of system is a totally ordered sequence of logical ticks

I At each tick, an arbitrary number of events (including 0)
occurs

I Event occurrences that happen at the same logical tick are
considered simultaneous

I Other events are ordered as their instances of occurrences

Uses of SSMs/Esterel

I Wristwatch
I Canonical example
I Reactive, synchronous, hard real-time

I Controllers
I Communication protocols

I Avionics
I Fuel control system
I Landing gear controller
I Other user interface tasks

I Processor components (cache controller, etc.)

I General hw design

Advantages of SSMs/Esterel

I Model of time gives programmer precise control

I Concurrency convenient for specifying control systems
I Completely deterministic

I Guaranteed: no need for locks, semaphores, etc.

I Finite-state language
I Easy to analyze
I Execution time predictable
I Much easier to verify formally

I Amenable to implementation in both hardware and software

Disadvantages of SSMs/Esterel

I Finite-state nature of the language limits flexibility
I No dynamic memory allocation
I No dynamic creation of processes

I Virtually nonexistent support for handling data (this changes
in v7)

I Must resort to some host language (e. g., C) for that

I Really suited for simple decision-dominated controllers

I Synchronous model of time can lead to overspecification
I Semantic challenges

I Avoiding causality violations often difficult
I Difficult to compile

I Limited number of users, tools, etc.

Fall 2007 EE 249 Slide 21



Introduction to Safe State Machines and Esterel
Esterel Language Overview

Esterel/SSM Pragmatics
Interfacing with the Environment

Property Verification

Signal emission + testing, pausing
Esterel’s model of time
Parallelism
Signal awaiting, looping
Preemption, exceptions, suspension

Overview

Introduction to Safe State Machines and Esterel

Esterel Language Overview
Signal emission + testing, pausing
Esterel’s model of time
Parallelism
Signal awaiting, looping
Preemption, exceptions, suspension

Esterel/SSM Pragmatics

Interfacing with the Environment

Property Verification Fall 2007 EE 249 Slide 22



Introduction to Safe State Machines and Esterel
Esterel Language Overview

Esterel/SSM Pragmatics
Interfacing with the Environment

Property Verification

Signal emission + testing, pausing
Esterel’s model of time
Parallelism
Signal awaiting, looping
Preemption, exceptions, suspension

Basic Esterel Statements

emit S

I Make signal S present in the current instant

I A signal is absent unless it is emitted

pause

I Stop and resume after the next cycle after the pause

present S then stmt1 else stmt2 end

I If signal S is present in the current instant, immediately run
stmt1, otherwise run stmt2

Esterel’s Model of Time

I The standard CS model (e.g., Java’s) is asynchronous
I Threads run at their own rate
I Synchronization is done (for example) through calls to wait()

and notify()

I Esterel’s model of time is synchronous like that used in
hardware. Threads march in lockstep to a global clock.

Time

Clock tick

Basic Esterel Statements

module EXAMPLE1:

output A, B, C;

emit A;

present A then

emit B

end;

pause;

emit C

end module

A
B

C

EXAMPLE1 makes signals A &
B present the first instant, C
present the second

Transformation of EXAMPLE1 into SSMs

Structural translation of Esterel into SSMs
Performed with KIEL tool, www. informatik. uni-kiel. de/ rtsys/ kiel/

Transformation of EXAMPLE1 into SSMs

After some optimizations

Transformation of EXAMPLE1 into SSMs

Final version

Signal Coherence Rules

I Each signal is only present or absent in a cycle, never both

I All writers run before any readers do

I Thus

present A else

emit A

end

is an erroneous program

Advantage of Synchrony

I Easy to control time

I Synchronization comes for free

I Speed of actual computation nearly
uncontrollable—Synchrony allows to specify function and
timing independently

I Makes for deterministic concurrency

I Explicit control of “before” “after” “at the same time”

Time Can Be Controlled Precisely

This guarantees every 60th S an M is emitted:

every 60 S do every invokes its body every 60th S
emit M emit takes no time (cycles)

end

S S S S S
M M

1 · · · 59 60 61 · · · 120

The || Operator

Groups of statements separated by || run concurrently and
terminate when all groups have terminated

[

emit A;

pause; emit B;

||

pause; emit C;

pause; emit D

];

emit E

A B
C D

E

Communication Is Instantaneous

A signal emitted in a cycle is visible immediately

[

pause; emit A;

pause; emit A

||

pause;

present A then

emit B end

]

A A
B

Bidirectional Communication

Processes can communicate back and forth in the same cycle

[

pause; emit A;

present B then

emit C end;

pause; emit A

||

pause;

present A then

emit B end

]

A A
B
C

Concurrency and Determinism

I Signals are the only way for concurrent processes to
communicate

I Esterel does have variables, but they cannot be shared

I Signal coherence rules ensure deterministic behavior

I Language semantics clearly defines who must communicate
with whom when

The Await Statement

I The await statement waits for a particular cycle

I await S waits for the next cycle in which S is present

[

emit A;

pause;

pause; emit A

||

await A; emit B

]

A A
B

The Await Statement

I await normally waits for a cycle before beginning to check

I await immediate also checks the initial cycle

[

emit A;

pause;

pause; emit A

||

await immediate A;

emit B

]

A A
B

Loops

I Esterel has an infinite loop statement
I Rule: loop body cannot terminate instantly

I Needs at least one pause, await, etc.
I Can’t do an infinite amount of work in a single cycle

loop

emit A;

pause;

pause;

emit B

end

A A A A
B B B

Loops and Synchronization
Instantaneous nature of loops plus await provide very powerful
synchronization mechanisms

loop

await 60 S;

emit M

end

S S S S S
M M

1 · · · 59 60 61 · · · 120

Preemption

I Often want to stop doing something and start doing
something else

I E.g., Ctrl-C in Unix: stop the currently-running program

I Esterel has many constructs for handling preemption

The Abort Statement

I Basic preemption mechanism

I General form:

abort

statement

when condition

I Runs statement to completion

I If condition ever holds, abort terminates immediately.

The Abort Statement

abort

pause;

pause;

emit A

when B;

emit C

A
C

Normal Termination

B
C

Aborted termination

B
C

Aborted termination;
emit A preempted

B A
C

Normal Termination
B not checked
in first cycle
(like await)

Strong vs. Weak Preemption

I Strong preemption:
I The body does not run when the preemption condition holds
I The previous example illustrated strong preemption

I Weak preemption:
I The body is allowed to run even when the preemption

condition holds, but is terminated thereafter
I weak abort implements this in Esterel

Strong vs. Weak Abort

Strong abortabort

pause;

pause;

emit A;

pause

when B;

emit C

B
C

emit A not allowed to run

Weak abortweak abort

pause;

pause;

emit A;

pause

when B;

emit C

A
B
C

emit A does run, body
terminated afterwards

Strong vs. Weak Preemption

I Important distinction

I Something cannot cause its own strong preemption

abort

pause;

emit A

when A

Erroneous!

weak abort

pause;

emit A

when A

Ok!

Fall 2007 EE 249 Slide 45



Introduction to Safe State Machines and Esterel
Esterel Language Overview

Esterel/SSM Pragmatics
Interfacing with the Environment

Property Verification

Signal emission + testing, pausing
Esterel’s model of time
Parallelism
Signal awaiting, looping
Preemption, exceptions, suspension

Nested Preemption
module RUNNER

input SECOND, METER, LAP;

output ... ;

every MORNING do

abort

loop

abort RUNSLOWLY when 15 SECOND;

abort

every STEP do

JUMP || BREATHE

end every

when 100 METER;

FULLSPEED

each LAP

when 2 LAP

end every

end module

Exceptions—The Trap Statement

I Esterel provides an exception facility for weak preemption

I Interacts nicely with concurrency

I Rule: outermost trap takes precedence

The Trap Statement

trap T in

[

pause;

emit A;

pause;

exit T

||

await B;

emit C

]

end trap;

emit D

A D Normal termination
from first process

A
B
C D emit C also runs

A B
C
D

Second process
allowed to run even
though first process
has exited

Nested Traps

trap T1 in

trap T2 in

[

exit T1

||

exit T2

]

end;

emit A

end;

emit B

I Outer trap takes precedence; control
transferred directly to the outer trap
statement.

I emit A not allowed to run.

B

Combining Abortion and Exceptions

trap HEARTATTACK in

abort

loop

abort RUNSLOWLY when 15 SECOND;

abort

every STEP do

JUMP || BREATHE || CHECKHEART

end every

when 100 METER;

FULLSPEED

each LAP

when 2 LAP

handle HEARTATTACK do

GOTOHOSPITAL

end trap

Parallelism
Signal awaiting, looping
Preemption, exceptions, suspension

The Suspend Statement

I Preemption (abort, trap) terminate something, but what if
you want to pause it?

I Like the POSIX Ctrl-Z

I Esterel’s suspend statement pauses the execution of a group
of statements

I Only strong preemption: statement does not run when
condition holds

Parallelism
Signal awaiting, looping
Preemption, exceptions, suspension

The Suspend Statement
suspend

loop

emit A;

pause;

pause

end

when B

A A B A B A

B delays emission
of A by one cycle

B prevents A
from being emitted here;
resumed next cycle

Overview

Introduction to Safe State Machines and Esterel

Esterel Language Overview

Esterel/SSM Pragmatics
People Counter Example
Vending Machine Example
Tail Lights Example
Traffic-Light Controller Example

Interfacing with the Environment

Property Verification

People Counter Example

Construct an Esterel program that counts the number of people in a
room.

I People enter the room from one door with a photocell that changes
from 0 to 1 when the light is interrupted, and leave from a second
door with a similar photocell. These inputs may be “1” for more
than one clock cycle. It is assumed that one continuous sequence of
1’s corresponds to a single person passing the photocell. The two
photocell inputs are called ENTER and LEAVE.

I There are two outputs: EMPTY and FULL, which are present when
the room is empty and contains three people respectively.

Source: Mano, Digital Design, 1984, p. 336

Overall Structure

ENTER

LEAVE

Conditioner

Conditioner
Counter

EMPTY

FULL

ADD

SUB

Conditioner detects rising edges of signal from photocell.
Counter tracks number of people in the room.

Implementing & Testing the Conditioner

module CONDITIONER:

input A;

output Y;

loop

await A; emit Y;

await [not A];

end

end module

Fall 2007 EE 249 Slide 56



Introduction to Safe State Machines and Esterel
Esterel Language Overview

Esterel/SSM Pragmatics
Interfacing with the Environment

Property Verification

People Counter Example
Vending Machine Example
Tail Lights Example
Traffic-Light Controller Example

Implementing & Testing the Conditioner

module CONDITIONER:

input A;

output Y;

loop

await A; emit Y;

await [not A];

end

end module

CONDITIONER> ;

--- Output:

CONDITIONER> A; # Rising edge

--- Output: Y

CONDITIONER> A; # Doesn’t generate a pulse

--- Output:

CONDITIONER> ; # Doesn’t generate a pulse

--- Output:

CONDITIONER> ; # Reset

--- Output:

CONDITIONER> A; # Another rising edge

--- Output: Y

CONDITIONER> ;

--- Output:

CONDITIONER> A;

--- Output: Y

Implementing & Testing the Counter: First Try
module COUNTER:

input ADD, SUB;

output FULL, EMPTY;

var count := 0 : integer in

loop

present ADD then

if count < 3 then

count := count + 1

end end;

present SUB then

if count > 0 then

count := count - 1

end end;

if count = 0 then emit EMPTY end;

if count = 3 then emit FULL end;

pause

end

end

end module

Implementing & Testing the Counter: First Try

COUNTER> ;

--- Output: EMPTY

COUNTER> ADD SUB;

--- Output: EMPTY

COUNTER> ADD;

--- Output:

COUNTER> SUB;

--- Output: EMPTY

COUNTER> ADD;

--- Output:

COUNTER> ADD;

--- Output:

COUNTER> ADD;

--- Output: FULL

COUNTER> ADD SUB;

--- Output: # Oops!

Implementing & Testing the Counter: Second Try
module COUNTER:

input ADD, SUB;

output FULL, EMPTY;

var c := 0 : integer in

loop

present ADD then

present SUB else

if c < 3 then c := c + 1 end end

else

present SUB then

if c > 0 then c := c - 1 end end

end;

if c = 0 then emit EMPTY end;

if c = 3 then emit FULL end;

pause

end

end

end module

Implementing & Testing the Counter: Second Try
COUNTER> ;

--- Output: EMPTY

COUNTER> ADD SUB;

--- Output: EMPTY

COUNTER> ADD SUB;

--- Output: EMPTY

COUNTER> ADD;

--- Output:

COUNTER> ADD;

--- Output:

COUNTER> ADD;

--- Output: FULL

COUNTER> ADD SUB;

--- Output: FULL # Working

COUNTER> ADD SUB;

--- Output: FULL

COUNTER> SUB;

--- Output:

COUNTER> SUB;

--- Output:

COUNTER> SUB;

--- Output: EMPTY

COUNTER> SUB;

--- Output: EMPTY

Assembling the People Counter

module PEOPLECOUNTER:

input ENTER, LEAVE;

output EMPTY, FULL;

signal ADD, SUB in

run CONDITIONER[signal ENTER / A, ADD / Y]

||

run CONDITIONER[signal LEAVE / A, SUB / Y]

||

run COUNTER

end

end module

Vending Machine Example

Design a vending machine controller that dispenses gum once.

I Two inputs, N and D, are present when a nickel and dime have
been inserted.

N = D =

I A single output, GUM, should be present for a single cycle
when the machine has been given fifteen cents.

GUM =

I No change is returned.

Source: Katz, Contemporary Logic Design, 1994, p. 389

Vending Machine Solution—Variable Encoding
module VENDING:

input N, D;

output GUM;

loop

var m := 0 : integer in

trap WAIT in

loop

present N then m := m + 5; end;

present D then m := m + 10; end;

if m >= 15 then exit WAIT end;

pause

end loop

end trap;

emit GUM; pause

end

end

end module

Alternative Solution—State Encoding

loop

await

case immediate N do await

case N do await

case N do nothing

case immediate D do nothing

end

case immediate D do nothing

end

case immediate D do await

case immediate N do nothing

case D do nothing

end

end;

emit GUM; pause

end

Tail Lights Example

Construct an Esterel program that controls the turn signals of a
1965 Ford Thunderbird.

Source: Wakerly, Digital Design Principles & Practices, 2ed, 1994, p. 550

Tail Light Behavior

Tail Lights

I There are three inputs, which initiate the sequences: LEFT,
RIGHT, and HAZ

I Six outputs: LA, LB, LC, RA, RB, and RC

I The flashing sequence is

LC LB LA step RA RB RC

1

2

3

4

A Single Tail Light

module LIGHTS:

output A, B, C;

loop

emit A; pause;

emit A; emit B; pause;

emit A; emit B; emit C; pause;

pause

end

end module

The T-Bird Controller: Esterel Version
input LEFT, RIGHT, HAZ;

output LA, LB, LC, RA, RB, RC;

loop

await

case immediate HAZ do

abort

run LIGHTS[signal LA/A, LB/B, LC/C]

||

run LIGHTS[signal RA/A, RB/B, RC/C]

when [not HAZ]

case immediate LEFT do

abort run LIGHTS[signal LA/A, LB/B, LC/C]

when [not LEFT]

case immediate RIGHT do

abort run Lights[signal RA/A, RB/B, RC/C]

when [not RIGHT]

end

end

The T-Bird Controller: SSM Version

input LEFT, RIGHT, HAZ;

output LA, LB, LC, RA, RB, RC;

Comments on the T-Bird

I This solution uses Esterel’s innate ability to control the
execution of processes, producing succinct easy-to-understand
source but a somewhat larger executable.

I An alternative: Use signals to control the execution of two
processes, one for the left lights, one for the right.

I A challenge: Synchronizing hazards.

I Most communication signals can be either level- or
edge-sensitive.

I Control can be done explicitly, or implicitly through signals.

Traffic-Light Controller Example

C

C
Control a traffic light at the intersection of
a busy highway and a farm road.
Source: Mead and Conway, Introduction to VLSI

Systems, 1980, p. 85.

I Normally, the highway light is green
I If a sensor detects a car on the farm road:

I The highway light turns yellow then red.
I The farm road light then turns green until there are no cars or

after a long timeout.
I Then, the farm road light turns yellow then red, and the

highway light returns to green.
I Inputs: The car sensor C, a short timeout signal S, and a long

timeout signal L.
I Outputs: A timer start signal R, and the colors of the highway

and farm road lights HG, HY, HR, FG, FY, and FR.

The Traffic Light Controller (SSM)

input C, SEC;

output HG, HY, HR,

FG, FY, FR;

input R, SEC;

output L, S;

input C, L, S;

output R, HG, HY, HR,

FG, FY, FR;

The Traffic Light Controller (Esterel)

module TLC:

input C, SEC;

output HG, HY, HR,

FG, FY, FR;

signal S, L, S in

run TIMER

||

run FSM

end

end module

module TIMER:

input R, SEC;

output L, S;

loop

weak abort

await 3 SEC;

[

sustain S

||

await 5 SEC;

sustain L

]

when R;

end

end module

module FSM:

input C, L, S;

output R, HG, HY, HR,

FG, FY, FR;

loop

emit HG; emit FR;

emit R;

await [C and L];

emit HY; emit R;

await S;

emit HR; emit FG;

emit R;

await [(not C) or L];

emit FY; emit R;

await S;

end

end module

Esterel Language Overview

Interfacing with the Environment

I At some point, our reactive system must control real-world
entities

I There are usually different options for the interface—differing
in

I Ease of use
I Ease of making mistakes!

I Example: External device that can be ON or OFF
I Options:

1. Single pure signal
2. Two pure signals
3. Boolean valued signal

Valued Signals

I Beside the status present or absent, a signal can have an
additional value.

I Valued signals are declared with a certain type

I output S: integer declares an output signal of type integer

I emit S(15) makes signal S present and assigns it the value
15

I Value of signal S can be tested by ?S

I The value is persistent across logical ticks

I To preserve determinism, only one signal value per tick allowed

Valued Signals

Single valued signal:

I Only one statement can emit signal per instant

Combined valued signal:

I Multiple emitters allowed

I Indicated with combine keyword

I Are combined with (commutative and associative) binary
operator

I boolean: combination function can be and or or

I integer, float, double: can use + or ∗

Variables

I . . . are assignable objects with name and type

I . . . similar rules as for signals (regarding placement, scoping)

I Value is undefined until first assignment

var

X : double,

Count := ? Distance : integer,

Deadline : Time

in

p

end var

I Must declare type individually for each variable
I var X, Y integer is incorrect!

Different Modes of Motor Control

Option 1: Single pure signal

I Motor is running in every instant
which has the MOTOR signal present

Pro:

I Minimal number of signals

Con:

I High number of signal emissions
(signal is emitted in every instant
where the motor is on)—may be
unnecessary run-time overhead

I Somewhat heavy/unintuitive
representation

input BUMPER;

output MOTOR;

abort

sustain MOTOR

when BUMPER

Different Modes of Motor Control

Option 2: Two pure signals

I Motor is switched on with signal
MOTOR ON present

I Motor is switched off with signal
MOTOR OFF present

I If neither MOTOR ON or MOTOR OFF is
present, motor keeps its previous state

input BUMPER;

output MOTOR_ON,

MOTOR_OFF;

emit MOTOR_ON;

await BUMPER;

emit MOTOR_OFF;

Pro:

I Signal emissions truly indicate significant change of external state

I Simple representation in Esterel

Con:

I No way to control inconsistent outputs

I No memory - cannot check in retrospect which signal was emitted

Inconsistent Outputs

I Problem with MOTOR ON and MOTOR OFF: undefined behavior
with both signals present

I Can address this at host-language level

I Can (and should) also address this at Esterel-level:

present BUMPER else

emit MOTOR_ON;

await BUMPER

end present;

emit MOTOR_OFF

||

await immediate MOTOR_ON and MOTOR_OFF;

exit INTERNAL_ERROR

Valued Signal for Motor Control

Option 3: Boolean valued signal

I Merge pure signals MOTOR ON and
MOTOR OFF into one valued signal
MOTOR

I Motor is switched on if every
emit-statement in that instant
emits true

I Here: In case of conflicting
outputs, motor stays switched off

input BUMPER;

output MOTOR

combine BOOLEAN

with and;

emit MOTOR(true);

await immediate BUMPER;

emit MOTOR(false);

Esterel Signal Types
Option 1: Single pure signal
Option 2: Two pure signals

Valued Signal for Motor Control
Option 3 contd.
Pro:

I Again only one signal for motor control

I Explicit control of behavior for inconsistent outputs

I Valued signal has memory—can be polled in later instances,
after emission

I Easy extension to finer speed control

Con:

I Inconsistent outputs are handled deterministically—but are
not any more detected and made explicit

I For certain classes of analyses/formal methods that we may
wish to apply, valued signals are more difficult to handle than
pure signals

Fall 2007 EE 249 Slide 85



Introduction to Safe State Machines and Esterel
Esterel Language Overview

Esterel/SSM Pragmatics
Interfacing with the Environment

Property Verification

Esterel Signal Types
Option 1: Single pure signal
Option 2: Two pure signals
Option 3: Boolean valued signal

Events vs. State
I Excessive signal emissions

I make the behavior difficult to understand
I cause overhead if fed to the external environment

I State:
I “Robot is turning left”
I “Motor is on”
I Esterel:

I waiting for some signal
I terminated thread
I value of valued signal

I Event:
I Change of State
I “Turn motor on”
I Esterel:

I emit pure signal
I change value of signal

Overview

Introduction to Safe State Machines and Esterel

Esterel Language Overview

Esterel/SSM Pragmatics

Interfacing with the Environment

Property Verification
Introduction
Example: ABRO

Property Verification

I One advantage of formal foundation of synchronous model:
Ability to formally verify certain properties

I Can conveniently specify properties using observers, using the
familiar SSM/Esterel formalism

I Observers scan for
I Always type properties (must always be fulfilled)
I Never type properties (should never occur)

I Verifier, based on model checking, is included in Esterel Studio

Fall 2007 EE 249 Slide 88



Introduction to Safe State Machines and Esterel
Esterel Language Overview

Esterel/SSM Pragmatics
Interfacing with the Environment

Property Verification

Introduction
Example: ABRO

Example: ABRO
Property P1:
O cannot be emitted if B has
not been received since the
last occurrence of R

Observer for P1:

Screenshot of Esterel-Studio Verifier

Example: ABRO
Property P2:
O is never emitted twice since
the last occurrence of R

Observer for P2:

Summary I

I Classical real-time languages include specific notions of
physical time—however, they do not achieve complete
determinism this way

I Synchronous languages replace notion of physical time with
notion of order, considering only simultaneity and precedence
of events—this is the multiform notion of time

I The Write-Things-Once principle aids to make representations
compact, and to ease modifications. For state machines,
WTO is achieved by adding hierarchy

Summary II

I Esterel is an imperative, control-oriented synchronous
language

I Synchronous model of time, as used by SSMs
I Time divided into sequence of discrete ticks
I Instructions either run and terminate in the same tick or

explicitly in later ticks

I Idea of signals and broadcast
I “Variables” that take exactly one value each tick and don’t

persist
I Coherence rule: all writers run before any readers

Summary III

I Esterel and SSMs are high-level descriptions—however, there
are still several options to express the same behavior

I May e. g.alternatively use state encoding or variable encoding
to memorize control state across logical ticks

I Can use macro facility to modularize description

I Care should be taken to select a suitable interface with the
environment—single pure signals, two pure signals, or Boolean
valued signal

I The formally founded semantics of Esterel allows to perform
formal verification (for more, should attend additional class,
e. g.“Verification of Concurrent Programs”)

To Go Further

I Nicolas Halbwachs, Synchronous programming of reactive systems,
a tutorial and commented bibliography, Tenth International
Conference on Computer-Aided Verification, CAV’98 Vancouver
(B.C.), LNCS 1427, Springer Verlag, June 1998, http:
//www-verimag.imag.fr/~halbwach/cav98tutorial.html

I Gérard Berry, The Foundations of Esterel, Proof, Language and
Interaction: Essays in Honour of Robin Milner, G. Plotkin, C.
Stirling and M. Tofte, editors, MIT Press, Foundations of
Computing Series, 2000, ftp:
//ftp.esterel.org/esterel/pub/papers/foundations.ps

I Esterel Web, http://www-sop.inria.fr/esterel.org/

I Home page of Esterel Technologies,
http://www.esterel-technologies.com/v3/

